

0.7 ジュニア版 記述統計量

モデルをもった回帰分析に進む前に、平均や分散から始まって、データの性質を知るための様々な値を 1 から求めてみましょう。すでにコマンドとして組み込まれているものもありますが、それらをあらためてプログラムするということはその仕組みを実感すると同時に、今後進んでいくプログラミングの土台の部分となります。(下では、もうすでに組み込み関数として存在する関数を 1 から作成するので、名前を区別するために敢えて日本語名をつけます。世界共通語のプログラムの通常関数名は英語の略称が望ましいことは言うまでもありません。)

では、これまでと同じようにプログラムエディター上にデータをカット&ペーストまたは手で打ち込む方法でデータを入力して、それを利用して各種の統計の値を調べてみましょう。ただし、組み込み関数は極力使わないものとします。

合計

プログラム

```
new;
cls;
let data[10,3]=
1  3 10
2  7  9
3  8  8
4  2  7
5  1  6
6  4  5
7 10  4
8  9  3
9  6  2
10 5  1
;
print goukei(data);

proc goukei(x);
  local n,i,sum;
  n=rows(x);
  sum=0;
  i=1;
```

```

do while i<=n;
    sum=sum+x[i,.];
    i=i+1;
end;
retp(sum');
endp;

```

画面表示

```

55.000000
55.000000
55.000000

```

プログラム各行の説明

- 1 行目 メモリを初期化する
- 2 行目 画面をクリアする
- 3 ~ 14 行目 変数 **data** に 10 × 3 のディメンションでセミコロンまでのデータを入れる
- 15 行目 自作関数 **goukei** を呼び出して、その中に変数 **data** を代入して、その答えを画面表示させる
- 17 ~ 27 行目 **proc** から **endp**;までが **goukei** という名前でインプット **x** の関数
- 18 行目 この関数の中だけで使われるローカル変数は **n,i,sum** の3つ
- 19 行目 組込み関数 **rows** を用いて変数 **x** の行数を調べて、それを変数 **n** に入れる
- 20 行目 今から足しあわせていく変数 **sum** の中に0を入れておく
- 21 行目 **do** ループに入る前にそのカウンタ **i** を1に設定しておく
- 22 行目 カウンタ **i** が **n** 以下「の間は」**end; until** までのことを繰り返す
- 23 行目 変数 **sum** と変数 **x** の **i** 行目を足し合わせて、あらためて変数 **sum** に入れる
(**x[i,.]**のドットマークは「すべての」列という意味)
- 24 行目 カウンタ **i** を1増やして、あらためて **i** とする
- 25 行目 **do** ループの終わり(ここまでを繰り返す)
- 26 行目 変数 **sum** を転置させたものをこの関数のリターンとして外部に返す
(**GAUSS** の内部関数と同じ出力形式にするため転置させただけである)
- 27 行目 この **procedure** の終わり(ここまでが1つの **procedure**)

上のプログラムではデータの部分を手で入力するか、または Copy&Paste で貼りつけたものを利用します。後半に置いている **goukei** という関数を通常の組込み関数と同じように呼び出してその中に変数を入れて **print** 命令で画面表示させています。

プログラム中のデータと同じ次のような 10 行 3 列の 10 × 3 の配列の行列を考えよう。

$$data = \begin{bmatrix} & \text{1 列} & \text{2 列} & \text{3 列} \\ \text{1 行} & 1 & 3 & 10 \\ \text{2 行} & 2 & 7 & 9 \\ \text{3 行} & 3 & 8 & 8 \\ \text{4 行} & 4 & 2 & 7 \\ \text{5 行} & 5 & 1 & 6 \\ \text{6 行} & 6 & 4 & 5 \\ \text{7 行} & 7 & 10 & 4 \\ \text{8 行} & 8 & 9 & 3 \\ \text{9 行} & 9 & 6 & 2 \\ \text{10 行} & 10 & 5 & 1 \end{bmatrix}$$

GAUSS では配列は

変数[行 , 列]

と表される。例えば data[5,3]は data の 5 行 3 列目の成分のことで、この場合 6 になる。同様に、data[10,1]は data の 10 行 1 列目の成分で 10 になっている。また、列または行をワイルドカードにすることも可能である。例えば、data[1,.]は data の 1 行目すべてを表し、この場合{ 1 3 10}のことである。同様に、data[10,.]は 10 行目すべてで{10 5 1}のことである。また列方向に考えると data[:,1]は 1 列目すべてのことで{1,2,3,4,5,6,7,8,9,10}のことである。これを利用して、カウンタ i を 1 から 1 つずつ増して 10 までを考えると

$$\begin{bmatrix} & \text{1 列} & \text{2 列} & \text{3 列} \\ i & 1 & 3 & 10 \\ & 2 & 7 & 9 \\ & 3 & 8 & 8 \\ & 4 & 2 & 7 \\ & 5 & 1 & 6 \\ & 6 & 4 & 5 \\ & 7 & 10 & 4 \\ & 8 & 9 & 3 \\ & 9 & 6 & 2 \\ & 10 & 5 & 1 \end{bmatrix} \rightarrow \begin{bmatrix} & \text{1 列} & \text{2 列} & \text{3 列} \\ i & 1 & 3 & 10 \\ & 2 & 7 & 9 \\ & 3 & 8 & 8 \\ & 4 & 2 & 7 \\ & 5 & 1 & 6 \\ & 6 & 4 & 5 \\ & 7 & 10 & 4 \\ & 8 & 9 & 3 \\ & 9 & 6 & 2 \\ & 10 & 5 & 1 \end{bmatrix} \dots \rightarrow \dots \begin{bmatrix} & \text{1 列} & \text{2 列} & \text{3 列} \\ & 1 & 3 & 10 \\ i & 2 & 7 & 9 \\ & 3 & 8 & 8 \\ & 4 & 2 & 7 \\ & 5 & 1 & 6 \\ & 6 & 4 & 5 \\ & 7 & 10 & 4 \\ & 8 & 9 & 3 \\ & 9 & 6 & 2 \\ i & 10 & 5 & 1 \end{bmatrix}$$

というふうに i 行目を表す data[i,.]が data[1,.],data[2,.],...data[10,.]と移動していく。この

ことを利用して、`sum=0` としておいた変数 `sum` に 1 行目から順に最大行数の 10 行目までを足し合わせたものが最終的に `sum` となって、それをアウトプットリターンにしている。この場合、それぞれの列はそれぞれの列のところにある数と足し合わされていくので、列のことは気にする必要がない。すなわち、ドットのワイルドカードとして考えています。もう少し対応関係の例を示しておくと、

<code>data[1,1]</code>	行列変数 <code>data</code> の 1 行 1 列目
<code>data[10,1]</code>	行列変数 <code>data</code> の 10 行 1 列目
<code>data[. ,1]</code>	行列変数 <code>data</code> の（行はワイルドカードで）1 列目すべて
<code>data[1, .]</code>	行列変数 <code>data</code> の 1 行目すべて（列はワイルドカード）

のようになります。

平均

今度は上の列ごとの合計をもとにして少しだけ変更して平均を求めてみます。

プログラム

```
new;  
cls;  
let data[10,3]=  
1 3 10  
2 7 9  
3 8 8  
4 2 7  
5 1 6  
6 4 5  
7 10 4  
8 9 3  
9 6 2  
10 5 1  
;  
print heikin(data);
```

```
proc heikin(x);  
  local n,i,sum,m;  
  n=rows(x);  
  sum=0;  
  i=1;  
  do while i<=n;
```

```

        sum=sum+x[i,:];
        i=i+1;
    endo;
    m=sum/n;
    retp(m');
endp;

```

画面表示

```

5.5000000
5.5000000
5.5000000

```

プログラム各行の説明（前のプログラムとの相違点）

- 1 8 行目 ローカル変数は **n,i,sum** に加えて **m** の 4 つである
- 2 6 行目 変数 **sum** を行数 **n** で割ったものを **m** とする
- 2 7 行目 変数 **m** を転置させたものをこの関数のアウトプットリターンとして返す

平均の関数は合計の関数までのプログラムを事実上 1 箇所だけでできてしまいます。つまり、合計で行ごとに足し合わせてできた変数 **sum** を行数 **n** で割ってやるだけでよいわけです。このように、関数を少し変更するには、新たにその関数内部で使うローカル変数を **local** 命令の後に加えることを忘れないで、関数名と **retp()** の括弧の中のアウトプットリターンの変数を新しく計算した変数に変更するだけでできます。

注意事項

- 1) プログラムでは何も入っていない変数にいきなり何かを足したり引いたりできない。したがって、**sum** には必ず何か計算の支障にならない数を入れておく必要がある（この場合は 0 としている）。
- 2) この **sum** もカウンタの **i** も共にループに入る前に外側で初期値を設定する必要がある。ループの内部で設定すると毎回その初期値に戻ってしまって都合が悪くなる。
- 3) GAUSS の組込み関数では、次に続く計算のことを考えて、計算結果は列として出てくることが多いので、自作関数の場合もこれにならい最後に行を列に転置させている。

最大値

新しい概念であるソート関数を用いて、これまでどおり行列を列ごとに考えて、小さいものから大きいものものに並べ替えてみましょう。GAUSS では、ソートはすでに高速なものが組み込み関数としてあるので、そこから出発することが効率的です。その使い方は、

sortc(変数,列番号)

としてやると、「その列番号のところを基準にして」「行としてまとめて」データが小さいものから大きいものへとソートされます。技術的には、私たちは複数の列を考えていますから、行列を 1 列ずつ取り出して、i=1 から列数まで 1 列ごとに第 1 列目のソートを繰り返します。すなわち、

変数[. ,i]=sortc(変数[. ,i],1)

を do while ループで i を 1 つずつ動かしてそれぞれの列ごとにソートします。

プログラム

(これ以前は省略)

print saidai(data);

proc saidai(x);

local n,k,i,max;

n=rows(x);

k=cols(x);

i=1;

do while i<=k;

x[. ,i]=sortc(x[. ,i],1);

i=i+1;

endo;

max=x[n,.];

retp(max');

endp;

画面表示

10.000000

10.000000

10.000000

プログラム (procedure 部分) 各行の説明

1 ~ 12 行目 proc から endp まだが saidai という関数名でインプット変数が x の関数

- 2 行目 この関数の中だけで使われるローカル変数は、**n,k,i,max** の 4 つ
- 3 行目 組込み関数 **rows** を用いて変数 **x** の行数を調べて、それを変数 **n** に入れる
- 4 行目 組込み関数 **cols** を用いて変数 **x** の列数を調べて、それを変数 **k** に入れる
- 5 行目 ループのカウンタ **i** を 1 に設定しておく
- 6 ~ 9 行目 **do while** から **endo** までを **i <= k** であるあいだ繰り返す
- 7 行目 変数 **x** の **i** 列目からなる列ベクトルを (第 1 列目を基準にして) 小さいものから大きいものへソートして、あらためて変数 **x** の **i** 列目に入れる
- 8 行目 カウンタ **i** を 1 だけ増して、あらためて **i** に入れる
- 10 行目 変数 **x** の **n** 行目のすべてを変数 **max** に入れる
- 11 行目 変数 **max** を転置したものをこの関数のアウトプットリターンとして外部に返す

すこしめんどうかかもしれませんが、わからない場合には、

```
i=1;  
do while i<=k;  
    x[:,i]=sortc(x[:,i],1);  
    i=i+1;  
endo;
```

の 5 行をソートをするものと考えて今後利用してみてください。 **i** 列目だけを (1 列しかありませんが、第 1 列目を基準に) ソートして、それをもとの変数の **i** 列目に入れていきます。ループをまわさなければ、ただ第 1 列目だけに注目したソートになってしまいます。その他の列は、第 1 列めに行ごとに一括ソートされてめちゃくちゃになってしまいます。なお、**n** は変数 **x** の行数のことですから、ワイルドカードのドットを用いて、**x[n,.]** は変数 **x** の **n** 行目となります。これにより、列ごとにソートされた後の第 **n** 行目、すなわち最大値ばかりが入った行ベクトルが取り出せます。それを最終的に **GAUSS** の出力方法に合わせて、転置させてアウトプットリターンとしています。

最小値

列ごとの最大値を求めたことを発展させて、簡単に最小値も求められます。ソートした後の結果の **n** 行目ではなくて、最小値は第 1 行目になります。

プログラム

(これ以前は省略)

```
print saisho(data);
```

```
proc saisho(x);  
    local n,k,i,min;  
    n=rows(x);
```

```

k=cols(x);
i=1;
do while i<=k;
    x[:,i]=sortc(x[:,i],1);
    i=i+1;
end;
min=x[1,:];
retp(min');
endp;

```

画面表示

```

1.0000000
1.0000000
1.0000000

```

上のように、列ごとのソート結果の変数 x の 1 行目（列はワイルドカード）を min と置いて、それを GAUSS の出力に合わせて転置させてアウトプットリターンとしています。なお、上で扱った行列の列ごとの「合計」「平均」「最大」「最小」は、GAUSS の組込み関数

```

sumc(data);
meanc(data);
maxc(data);
minc(data);

```

と直接しても求められます。上の結果が同じになるのか自分で確認してみましょう。

範囲（レンジ）

最大値と最小値がわかっているならば、それらの差である

$$\text{レンジ} = \text{最大値} - \text{最小値}$$

も最大と最小のプログラムを合わせて少し改造すれば求まるはずです。

プログラム

```

( これ以前は省略 )
print hani(data);

proc hani(x);
    local n,k,i,min,max,range;
    n=rows(x);
    k=cols(x);

```



```

i=1;
do while i<=k;
    x[:,i]=sortc(x[:,i],1);
    i=i+1;
enddo;
min=x[1,:];
max=x[n,:];
range=max-min;
retp(range');
endp;

```

画面表示

```

9.0000000
9.0000000
9.0000000

```

プログラム各行の説明（前のプログラムとの相違点）

- 10 行目 ソート後の行列変数 x の 1 行目（列はワイルドカード）を変数 min に入れる
- 11 行目 ソート後の行列変数 x の n 行目（列はワイルドカード）を変数 max に入れる
- 12 行目 max から min を引いたものを変数 range に入れる
- 13 行目 変数 range を転置させたものをアウトプットリターンとして外部に返す

最大と最小のプログラムを両方一緒に行なって、最後にその差をとっています。もし以前のプログラムを変更してこのプログラムを書くには、**procedure** の最初の部分の名前を例えば **hani** で、その次の行でローカル変数に **max** と **min** の両方に加えて、新しく **range** があるか確認します。もちろん、この **procedure** を呼び出す場合も新しい **procedure** 名である **hani** の丸括弧の中に変数 **data** を入れる必要があります。

中位値

今度は平均と並んでよく使われる、データの並びの真ん中の順位にある値を求めてみます。ただし、データの個数が偶数の場合（例えば 10 個の場合）真ん中には数はデータがありませんからその前後の平均を計算します。そこで、If 分岐を導入してプログラムしてみましょう。つまり、データの個数が偶数の場合と奇数の場合に分岐させます。下のプログラムでは、その判断をするためにデータの個数 n を 2 で割った数を四捨五入した数と n 割る 2 が等しい場合（偶数に相当）とそうでない場合（奇数に相当）に分岐させています。

プログラム

```

（ これ以前は省略 ）
print chuichi(data);

```

```

proc chuichi(x);
    local n,k,i,med;
    n=rows(x);
    k=cols(x);
    i=1;
    do while i<=k;
        x[:,i]=sortc(x[:,i],1);
        i=i+1;
    endo;
    if n/2==round(n/2);
        med=(x[n/2,.] + x[n/2+1,])/2;
    else;
        med=x[(n+1)/2,];
    endif;
    retp(med');
endp;

```

画面表示

```

5.5000000
5.5000000
5.5000000

```

パーセント点

プログラム

(これ以前は省略)

```

print percent(data,0.1);
print percent(data,0.5);
print percent(data,0.9);

```

```

proc percent(x,a);
    local n,k,i,per;
    n=rows(x);
    k=cols(x);
    i=1;
    do while i<=k;
        x[:,i]=sortc(x[:,i],1);

```

```

        i=i+1;
    endo;
    per=x[round(n*a),.];
    retp(per);
endp;

```

画面表示

1.0000000	1.0000000	1.0000000
5.0000000	5.0000000	5.0000000
9.0000000	9.0000000	9.0000000

四分位点

プログラム

(これ以前は省略)

```

print "1st" shibuni(data,0.25);
print "2nd" shibuni(data,0.5);
print "3rd" shibuni(data,0.75);
e={0.25,0.5,0.75};
quantile(data,e);

```

```

proc shibuni(x,p);
    local n,k,i,a,b,f,q;
    n=rows(x);
    k=cols(x);
    i=1;
    do while i<=k;
        x[:,i]=sortc(x[:,i],1);
        i=i+1;
    endo;
    a=floor(n*p);
    f=n*p-a;
    q=x[a,.]+f*(x[a+1,.]-x[a,.]);
    retp(q);
endp;

```

画面表示

1st	2.5000000	2.5000000	2.5000000
2nd	5.0000000	5.0000000	5.0000000

3rd	7.5000000	7.5000000	7.5000000
	2.5000000	2.5000000	2.5000000
	5.0000000	5.0000000	5.0000000
	7.5000000	7.5000000	7.5000000

「中位値」「パーセント点」「四分位点」の違い

これら3つは、定義によっては、等しくなる。しかしながら、GAUSSをはじめとする統計量ソフトウェアでは、中位値と第2四分位点は通常一致しない。四分位計算では補正が行なわれているからだ。また、四捨五入した50%の点であるパーセント点とも異なる。

```
new;
cls;
rndseed 1000;
data=rndn(99,2);

print "median" chuichi(data)';
print " 50%" percentile(data,0.5)';
print " 1st" shibuni(data,0.25);
print " 2nd" shibuni(data,0.5);
print " 3rd" shibuni(data,0.75);
```

```
proc shibuni(x,p);
  local n,k,i,a,b,f,q;
  n=rows(x);
  k=cols(x);
  i=1;
  do while i<=k;
    x[,i]=sortc(x[,i],1);
    i=i+1;
  endo;
  a=floor(n*p);
  f=n*p-a;
  q=x[a,]+f*(x[a+1,]-x[a,]);
```

```

    retp(q);
endp;

proc chuichi(x);
    local n,k,half,i,med;
    n=rows(x);
    k=cols(x);
    half=n/2;
    i=1;
    do while i<=k;
        x[:,i]=sortc(x[:,i],1);
        i=i+1;
    endo;
    if half==round(half);
        med=(x[n/2,.] + x[n/2+1,])/2;
    else;
        med=x[n/2,];
    endif;
    retp(med');
endp;

```

```

proc percentile(x,p);
    local n,k,i,xp;
    n=rows(x);
    k=cols(x);
    i=1;
    do while i<=k;
        x[:,i]=sortc(x[:,i],1);
        i=i+1;
    endo;
    xp=x[round(p*n),.];
    retp(xp');
endp;

```

画面表示

median	-0.11799732	0.22035185
50%	-0.10196121	0.22575022

1st	-0.71336306	-0.46742059
2nd	-0.10997926	0.22305103
3rd	0.82544983	0.66333897

プログラムを見てわかるように、GAUSS の動きをそのまま忠実に再現する定義では、中位値（メディアン）は、データ数が奇数の場合には小さい方から並べなおして中間の番目の値、偶数の場合にはその前後 2 つの平均値である。一方、単純パーセント点は、データ数にパーセントの小数値をかけたものを四捨五入して整数化したものの番目の値である。最後に、四分位計算で求める第 2 四分位点は a と $a + 1$ の間に位置する実際のパーセント番目を x の a 番目と $a + 1$ 番目を比例配分するように補正した値にする。最初の 1 から 10 までのデータなのではこれらは一致することもあるが、上のように通常は一致しない。教科書によって、あるいはソフトウェアによって定義が異なるおそれがあるので、使う前に比較計算が必要である。上の四捨五入の単純パーセント点を、複雑な計量計算で、簡単化のために使うこともよくあることなのでどれが誤りでどれが正しいということではない。

分散

プログラム

```
new;
```

```
cls;
```

```
let data[10,3]=
```

```
1 3 10
```

```
2 7 9
```

```
3 8 8
```

```
4 2 7
```

```
5 1 6
```

```
6 4 5
```

```
7 10 4
```

```
8 9 3
```

```
9 6 2
```

```
10 5 1
```

```
;
```

```
print bunsan(data);
```

```
proc bunsan(x);
```

```
local n,sum,i,xbar,s2;
```

```
n=rows(x);
```

```

sum=0;
i=1;
do while i<=n;
    sum=sum+x[i,.];
    i=i+1;
endo;
xbar=sum/n;
s2=sumc((x-xbar)^2)/(n-1);
retp(s2);
endp;

```

画面表示

```

9.1666667
9.1666667
9.1666667

```

ただし、上の計算では $n-1$ で割るバージョンである。サンプルではなくてポピュレーションを考えなくてはならない場合やその他の理由により n で割るバージョンも多用される。GAUSS では常に上のような $n-1$ で割るバージョンの計算なので、それが不都合ならば適宜自分で上のような関数を作り $n-1$ で割るところを n で割るものにする必用がある。

変動係数

プログラム

(これ以前は省略)

```
print cv(data);
```

```

proc cv(x);
    local n,sum,i,xbar,s;
    n=rows(x);
    sum=0;
    i=1;
    do while i<=n;
        sum=sum+x[i,.];
        i=i+1;
    endo;
    xbar=sum/n;
    s=sqrt(sumc((x-xbar)^2)/(n-1));
    retp(s./xbar');
endp;

```

endp;

画面表示

0.55048188

0.55048188

0.55048188

標準化

プログラム

(これ以前は省略)

print standard(data);

proc standard(x);

local n,sum,i,xbar,s,z;

n=rows(x);

sum=0;

i=1;

do while i<=n;

sum=sum+x[i,.];

i=i+1;

endo;

xbar=sum/n;

s=sqrt(sumc((x-xbar)^2)/(n-1));

z=(x-xbar)/s';

retp(z);

endp;

画面表示

-1.4863011

-0.82572282

1.4863011

-1.1560120

0.49543369

1.1560120

-0.82572282

0.82572282

0.82572282

-0.49543369

-1.1560120

0.49543369

-0.16514456

-1.4863011

0.16514456

0.16514456

-0.49543369

-0.16514456

0.49543369

1.4863011

-0.49543369

0.82572282

1.1560120

-0.82572282

1.1560120	0.16514456	-1.1560120
1.4863011	-0.16514456	-1.4863011

歪度（スキューネス）

（ これ以前は省略 ）

```
print hizumi(data);
```

```
proc hizumi(x);
```

```
    local n,sum,i,xbar,s,z,sk;
```

```
    n=rows(x);
```

```
    sum=0;
```

```
    i=1;
```

```
    do while i<=n;
```

```
        sum=sum+x[i,.];
```

```
        i=i+1;
```

```
    endo;
```

```
    xbar=sum/n;
```

```
    s=sqrt(sumc((x-xbar)^2)/(n-1));
```

```
    z=(x-xbar)./s';
```

```
    sk=sumc((z)^3)/n;
```

```
    retp(sk);
```

```
endp;
```

画面表示

```
-4.4408921e-017
```

```
-4.4408921e-017
```

```
4.4408921e-017
```

尖度（クルトーシス）

プログラム

（ これ以前は省略 ）

```
print togari(data);
```

```

proc togari(x);
    local n,sum,i,xbar,s,z,kur;
    n=rows(x);
    sum=0;
    i=1;
    do while i<=n;
        sum=sum+x[i,.];
        i=i+1;
    endo;
    xbar=sum/n;
    s=sqrt(sumc((x-xbar)^2)/(n-1));
    z=(x-xbar)/s';
    kur=sumc((z)^4)/n;
    retp(kur);
endp;

```

画面表示

```

1.4383636
1.4383636
1.4383636

```

共分散

プログラム

(これ以前は省略)

```

print kyobun(data);

```

```

proc kyobun(x);
    local n,sum,i,xbar,cov;
    n=rows(x);
    sum=0;
    i=1;
    do while i<=n;
        sum=sum+x[i,.];
        i=i+1;
    endo;
    xbar=sum/n;

```

```

cov=(x-xbar)'(x-xbar)/(n-1);
retp(cov);
endp;

```

画面表示

```

9.1666667      2.3888889     -9.1666667
2.3888889      9.1666667     -2.3888889
-9.1666667     -2.3888889      9.1666667

```

相関係数

プログラム

(これ以前は省略)

```

print soukan(data);

```

```

proc soukan(x);
  local n,sum,i,xbar,cov,s,corr;
  n=rows(x);
  sum=0;
  i=1;
  do while i<=n;
    sum=sum+x[i,];
    i=i+1;
  endo;
  xbar=sum/n;
  cov=(x-xbar)'(x-xbar)/(n-1);
  s=sqrt(sumc((x-xbar)^2)/(n-1));
  corr=cov./(s.*s');
  retp(corr);
endp;

```

画面表示

```

1.0000000      0.2606060     -1.0000000
0.2606060      1.0000000     -0.2606060
-1.0000000     -0.2606060      1.0000000

```

以上、普段よく見かけるような単純な「記述統計量」でさえ、考え方や定義により微妙な違いが生じることがよくわかったと思います。プログラミングとは、これらの違いをそのまま受け入れるのではなくて、自分なりにカスタマイズして利用することにあります。

そうしたカスタマイズの延長上に、いくつかある条件のうち1つを緩めて、新しいあるいは高度な統計や計量の計算をすることがあります。

練習問題

【問1】

上の補正された四分位計算を使って、

$$(\text{四分位範囲}) = (\text{第3四分位点}) - (\text{第1四分位点})$$

となる四分位範囲を求めると同時に、標準偏差を $n-1$ で割る従来の分散を求める計算のルートをとったものとして表示して、両者を比較しよう。

【問2】

分散および共分散の関数を n で割るものに変更して、それぞれ `bunsan2` と `kyobun2` としてやり、それを呼び出す形で分散と共分散を表示してやろう。その際、共分散行列の対角成分が分散に一致していることを確認しよう。

【問3】

分散の計算で $n-1$ で割るものの方が、 n で割るものよりも若干大きくなることをまず確認しよう。いくつかの統計学または計量経済学の教科書を見て、なぜ $n-1$ で割るものと n で割るものがあるのかを調べて、他の人に説明できるようにしよう。